

1.1 Vectores geométricos

2/11/2019/

1 - 1.1.7 Sintaxis Básica

Haremos algunos cálculos sencillos para ir calentando.

```
→      log(20);  
(%o1)   log(20)
```

Es probable que necesitemos el valor numérico de $\log(20)$.

```
→      log(20),numer;  
(%o2)  2.995732273553991  
  
→      420/16000;  
(%o3)   $\frac{21}{800}$   
  
→      420/16000,numer;  
(%o4)  0.02625
```

Podemos utilizar la función float para el mismo resultado

```
→      float(log(20)); float(420/16000);  
(%o5)  2.995732273553991  
(%o6)  0.02625
```

El programa contiene una gran cantidad de funciones matemáticas básicas internas y entre ellas las trigonométricas:

```
→      sin(%pi/3);cos(%pi/3);tan(%pi/3);  
(%o7)   $\frac{\sqrt{3}}{2}$   
(%o8)   $\frac{1}{2}$   
(%o9)   $\sqrt{3}$ 
```

Ahora recurriremos a una de las facilidades que nos ofrece el programa para agrupar objetos matemáticos: las listas. Las listas se escriben entre corchetes y los objetos de la lista separados con comas.

```
→ [sin(%pi/3),cos(%pi/3),tan(%pi/3)];
(%o10) [ $\frac{\sqrt{3}}{2}$ ,  $\frac{1}{2}$ ,  $\sqrt{3}$ ]

→ [sin(%pi/3),cos(%pi/3),tan(%pi/3)],numer;
(%o11) [0.8660254037844386, 0.5000000000000001,
1.732050807568877]
```

Cuando necesitemos generar una lista por medio de alguna regla específica o formula usamos la función "makelist". La sintaxis es la siguiente:

```
→ makelist(exp(t·x),t,1,10);
(%o12) [ex, e2 x, e3 x, e4 x, e5 x, e6 x, e7 x, e8 x, e9 x, e10 x]
```

Podemos también aplicar una función a cada elemento de la lista, en este caso a cada elemento le aplicaremos la función ln(x). Para tal fin utilizaremos el comando map.

```
→ map(log,(makelist(exp(t·x),t,1,10)));
(%o13) [x, 2 x, 3 x, 4 x, 5 x, 6 x, 7 x, 8 x, 9 x, 10 x]
```

Si queremos, por ejemplo, sumar todos los elementos de la lista anterior utilizamos la función apply con la operación que queremos realizar. Aquí aprovecharemos para utilizar un atajo muy práctico que consiste en el uso del símbolo %, que toma la última salida del programa para ser usado en la instrucción siguiente, de esta manera nos evitamos volver a escribir toda la instrucción anterior. La sintaxis para todo esto es:

```
→ apply("+",%);
(%o14) 55 x
```

Podemos asignarle a una lista el nombre de una variable

```
→ L: [sin(%pi/3),log(3),sqrt(2),abs(x),exp(x^2)];
(L) [ $\frac{\sqrt{3}}{2}$ , log(3),  $\sqrt{2}$ , |x|, ex2]
```

De manera que para aislar elementos de una lista escribimos:

→ $L[1]; L[4];$

$$(\%o16) \quad \frac{\sqrt{3}}{2}$$

$$(\%o17) \quad |x|$$

Esto nos permite operar con sus elementos.

→ $L[2] \cdot (L[1] + L[4]) / L[5];$

$$(\%o18) \quad \log(3) e^{-x^2} \left(|x| + \frac{\sqrt{3}}{2} \right)$$

La primera utilidad que le podemos dar a las listas es que nos permite definir vectores. Si queremos que el programa interprete los vectores $a = (a_1, a_2, a_3)$, $b = (b_1, b_2, b_3)$, $c = (c_1, c_2, c_3)$, $v_0 = (0, 0, 0)$, los podemos escribir como listas:

→ $a:[a_1, a_2, a_3]; b:[b_1, b_2, b_3]; c:[c_1, c_2, c_3]; v_0:[0, 0, 0];$

$$(a) \quad [a_1, a_2, a_3]$$

$$(b) \quad [b_1, b_2, b_3]$$

$$(c) \quad [c_1, c_2, c_3]$$

$$(v_0) \quad [0, 0, 0]$$

Como vimos, las operaciones básicas sobre los vectores cumplen un conjunto de propiedades:

→ $a+b=b+a;$

$$(\%o23) \quad [b_1+a_1, b_2+a_2, b_3+a_3] = [b_1+a_1, b_2+a_2, b_3+a_3]$$

→ $(a+b)+c=a+(b+c);$

$$(\%o24) \quad [c_1+b_1+a_1, c_2+b_2+a_2, c_3+b_3+a_3] = [c_1+b_1+a_1, c_2+b_2+a_2, c_3+b_3+a_3]$$

→ $a+v_0=a;$

$$(\%o25) \quad [a_1, a_2, a_3] = [a_1, a_2, a_3]$$

→ $a-a=v_0;$

$$(\%o26) \quad [0, 0, 0] = [0, 0, 0]$$

→ $\alpha \cdot (a+b) = \alpha \cdot a + \alpha \cdot b, \text{factor};$

$$(\%o27) \quad [(b_1+a_1)\alpha, (b_2+a_2)\alpha, (b_3+a_3)\alpha] = [(b_1+a_1)\alpha, (b_2+a_2)\alpha, (b_3+a_3)\alpha]$$

→ $(\alpha + \beta) \cdot a = \alpha \cdot a + \beta \cdot a$, factor;
(%o28) $[a_1(\beta + \alpha), a_2(\beta + \alpha), a_3(\beta + \alpha)] = [a_1(\beta + \alpha), a_2(\beta + \alpha), a_3(\beta + \alpha)]$

Con el siguiente comando limpiamos la memoria y borramos todos los cálculos anteriores

→ kill(all)\$

FIN

2 - 1.2.8 Vectores en componentes

Con el programa de manipulación simbólica Maxima haremos algunos cálculos sencillos con vectores. Nuevamente recomendamos ver el apéndice 6.1 como introducción al programa.

Dados los vectores, en coordenadas cartesianas: $a = i + 2j + 3k$ y $b = 7i + 8j + 9k$

→ $a:[1,2,3];$
(a) $[1, 2, 3]$

→ $b:[7,8,9];$
(b) $[7, 8, 9]$

La multiplicación por escalares y suma es simple, si queremos calcular $\alpha a + \beta b$, escribimos:

→ $\alpha \cdot a + \beta \cdot b;$
(%o3) $[7\beta + \alpha, 8\beta + 2\alpha, 9\beta + 3\alpha]$

Para el producto escalar procedemos utilizando el operador punto, como se muestra a continuación.

→ $a.b;$
(%o4) 50

El cálculo de producto vectorial no es tan obvio, debemos cargar previamente la librería vect.

```

→      load(vect)$

→      express(a~b);
(%o6)  [-6,12,-6]

```

La norma de un vector, como ya vimos, es: $\sqrt{a \cdot a}$

```

→      sqrt(a.a);

(%o7)  √14

```

Si tenemos otro vector, digamos $c = -4i + 5j - 6k$, el producto triple: $a \cdot b \times c$ se calcula así:

```

→      c:[-4,5,-6];
(c)   [-4,5,-6]

→      a.express(b~c);
(%o9)  120

```

El ángulo entre los vectores a y b . En Maxima usamos la función $\text{acos}(x)$ para el arcocoseno(x). Consultar el manual del programa para ver el resto de las funciones trigonométricas

```

→      acos((a.b)/(sqrt(a.a).sqrt(b.b)));
(%o10)   $\text{acos}\left(\frac{50}{\sqrt{14} \sqrt{194}}\right)$ 

```

Seguramente lo queremos es el valor numérico, esto se hace agregando la función `float`. Con la siguiente sintaxis logramos el objetivo:

```

→      acos((a.b)/(sqrt(a.a).sqrt(b.b))),float;
(%o11)  0.2858867976945072

```

Para finalizar, podemos considerar el problema de la independencia lineal de vectores. Tomemos el conjunto de vectores del ejemplo 1.2.7, es decir, los vectores: $a = i+2j+3k$, $b = 4i+5j$ y $c = 3i+2j+k$

Disponemos de un comando que permite resolver sistemas de ecuaciones lineales, este comando se llama "linsolve". Podemos escribir el sistema de ecuaciones como una lista y luego resolver:

```

→      ecus:[α+4·β+3·γ=0,2·α+5·β+2·γ=0,3·α+γ=0];
(ecus) [3 γ+4 β+α=0,2 γ+5 β+2 α=0,γ+3 α=0]

→      linsolve(ecus,[α,β,γ]);
(%o13) [α=0,β=0,γ=0]

→      kill(all)$

```

FIN

3 - 1.3.4 Aplicaciones del álgebra vectorial

Consideremos el plano determinado por los puntos $P = (1, 1, 1)$, $Q = (2, 4, 6)$ y $R = (2, 2, 2)$. Para construir la ecuación del plano primero ingresamos los vectores posición de cada punto. También incorporaremos las librerías "vect" y "draw" que necesitaremos.

```

→      load(vect)$ load(draw)$

→      P:[1,1,1];
(P)    [1,1,1]

→      Q:[2,4,6];
(Q)    [2,4,6]

→      R:[2,2,2];
(R)    [2,2,2]

```

Podemos verificar que los vectores estén en el mismo plano simplemente calculando el triple producto vectorial: $P \cdot (Q \times R)$ entre ellos. Sabemos que si es nulo es porque los vectores son coplanares.

```

→      P.express(Q~R);
(%o6)  0

```

Necesitamos ahora calcular los vectores que van del punto P al punto Q y del punto P al punto R , es decir, los vectores:

$$PQ = Q - P \text{ y } PR = R - P$$

```
→ PQ:=Q-P;
(PQ) [1,3,5]

→ PR:=R-P;
(PR) [1,1,1]
```

Un vector normal al plano será sencillamente el vector $\mathbf{N} = \mathbf{PQ} \times \mathbf{PR}$:

```
→ N:=cross(PQ,PR);
(N) [-2,4,-2]
```

Podemos escoger cualquiera de los vectores originales, en este caso al vector \mathbf{P} , para escribir la ecuación del plano: $\mathbf{N} \cdot \mathbf{r} = \mathbf{N} \cdot \mathbf{P}$.

```
→ r:=[x,y,z];
(r) [x,y,z]


→ plano:=N.r=N.P;
(plano) -2 z+4 y-2 x=0
```

Probemos graficar el plano. Para saber mas sobre las diferentes opciones que incorpora Maxima para hacer gráficas, en dos o tres dimensiones, es recomendable consultar el manual de usuario. Utilizaremos el comando "wxdraw3d" para hacer que la figura aparezca embebida dentro de nuestra hoja de trabajo.

Es recomendable consultar también las funciones "draw" y "gr3d".

→ `wxdraw3d(implicit(plano,x,-1,1,y,-1,1,z,-1,1));`

(%t12)



(%o12)

4 - 1.4.6 Álgebra vectorial con índices

A fines prácticos, las transformaciones de coordenadas del tipo rotaciones o reflexiones es de utilidad representarlas por matrices. Por ejemplo, las rotaciones alrededor del eje z se puede representar como:

→ `Rz:matrix([\cos(\theta),\sin(\theta),0],[-\sin(\theta),\cos(\theta),0],[0,0,1]);`

$$(Rz) \begin{bmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

De manera que un vector a transformará bajo esta rotación en un nuevo vector a'

→ `a:[a1,a2,a3];`

(a) `[a1, a2, a3]`

→ `Rz.a;`

$$(%o15) \begin{bmatrix} a2 \sin(\theta) + a1 \cos(\theta) \\ a2 \cos(\theta) - a1 \sin(\theta) \\ a3 \end{bmatrix}$$

Si la rotación se hace al rededor del eje x o y las matrices correspondientes son las siguientes:

→ Rx:matrix([1,0,0],[0,cos(θ),-sin(θ)],[0,sin(θ),cos(θ)]);

$$(Rx) \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(\theta) & -\sin(\theta) \\ 0 & \sin(\theta) & \cos(\theta) \end{bmatrix}$$

→ Ry:matrix([cos(θ),0,sin(θ)],[0,1,0],[-sin(θ),0,cos(θ)]);

$$(Ry) \begin{bmatrix} \cos(\theta) & 0 & \sin(\theta) \\ 0 & 1 & 0 \\ -\sin(\theta) & 0 & \cos(\theta) \end{bmatrix}$$

Notemos que el determinante de estas matrices es:

→ determinant(Rx),trigsimp;

$$(\%o18) \quad \sin(\theta)^2 + \cos(\theta)^2$$

→ trigrat(%);

$$(\%o19) \quad 1$$

Si la rotación es en un ángulo de $\theta = 60^\circ$ entonces:

→ sublis([θ=%pi/3], Rx);

$$(\%o20) \begin{bmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{2} & -\frac{\sqrt{3}}{2} \\ 0 & \frac{\sqrt{3}}{2} & \frac{1}{2} \end{bmatrix}$$

Si el eje de rotación se define a través de un vector unitario, digamos, $\hat{u} = ux\ i + uy\ j + uz\ k$, la matriz de rotación para un ángulo θ es:

→ t:1-cos(θ)\$

→ R:matrix([cos(θ)+ux^2·t,ux·uy·t-uz·sin(θ),ux·uz·t+uy·sin(θ)],[uy·ux·t+uz·sin(θ),cos(θ)+uy^2·t,uy·uz·t-ux·sin(θ)],[ux·uz·t-uy·sin(θ),uy·uz·t-ux·sin(θ),cos(θ)+uz^2·t])

$$(R) \begin{bmatrix} \cos(\theta) + ux^2 (1 - \cos(\theta)) & ux \, uy \, (1 - \cos(\theta)) - uz \sin(\theta) & uy \sin(\theta) + ux \, uz \, (1 - \cos(\theta)) \\ uz \sin(\theta) + ux \, uy \, (1 - \cos(\theta)) & \cos(\theta) + uy^2 (1 - \cos(\theta)) & uy \, uz \, (1 - \cos(\theta)) - ux \sin(\theta) \\ ux \, uz \, (1 - \cos(\theta)) - uy \sin(\theta) & ux \sin(\theta) + uy \, uz \, (1 - \cos(\theta)) & \cos(\theta) + uz^2 (1 - \cos(\theta)) \end{bmatrix}$$

Por ejemplo, si el eje de rotación coincide con el vector $\hat{u} = 1/\sqrt{3} i + j/\sqrt{3} j + k/\sqrt{3} k$ y la rotación es en un ángulo de $\theta = 60^\circ$, entonces:

→ [ux,uy,uz]:1/sqrt(3)·[1,1,1]; θ:%pi/3\$

(%o23) $[\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}]$

Al evaluar con el comando "ev" podemos ver como queda la matriz R con estos valores definidos

→ Ru:ev(R);

(Ru)

$$\begin{bmatrix} \frac{2}{3} & -\frac{1}{3} & \frac{2}{3} \\ \frac{2}{3} & \frac{2}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{2}{3} & \frac{2}{3} \end{bmatrix}$$

De manera que un vector, por ejemplo, $a = i + 2j + 3k$, transformará bajo esta rotación de la manera siguiente:

→ a:[1,2,3];

(a) [1,2,3]

→ Ru.a;

(%o27)

$$\begin{bmatrix} 2 \\ 1 \\ 3 \end{bmatrix}$$

Por otro lado, una reflexión en el plano xy es más sencilla de representar porque es la matriz:

→ Re:=matrix([-1,0,0],[0,1,0],[0,0,1]);
 (Re)
$$\begin{bmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

De manera que bajo ésta reflexión el vector transforma como:

→ Re.a;
 (%o29)
$$\begin{bmatrix} -1 \\ 2 \\ 3 \end{bmatrix}$$

 → kill(all)\$

FIN

5 - 1.5.8 Un comienzo a la derivación e integración de vectores

Maxima nos permite hacer cálculos con el operador ∇ en coordenadas cartesianas, más adelante veremos que también es posible en otros sistemas de coordenadas. Los operadores que se pueden expresar son:
 grad(gradiente), div(divergencia), curl(rotacional), laplacian(laplaciano).
 Debemos hacer uso de la librería "vect" para poder utilizar los operadores diferenciales.

Por otro lado, se debe usar también la función "express(expr)" para transformar los nombres de los operadores diferenciales en expresiones que contienen derivadas parciales y con la función ev evaluamos la expresión.

Veamos como se hace si queremos calcular el gradiente de la siguiente función:

→ load(vect)\$
 → f: (x^2+y^2)/(x^2+y^2+z^2)^(1/2);
 (f)
$$\frac{y^2 + x^2}{\sqrt{z^2 + y^2 + x^2}}$$

→ **grad(f);**
 (%o3) $\text{grad}\left(\frac{y^2+x^2}{\sqrt{z^2+y^2+x^2}}\right)$

Estaremos haciendo uso del símbolo % que representa la última expresión de salida, de esta manera nos ahorraremos estar escribiendo nuevamente el último resultado en la nueva linea de entrada.

→ **express(%);**
 (%o4) $\left[\frac{\frac{d}{dx} \frac{y^2+x^2}{\sqrt{z^2+y^2+x^2}}, \frac{d}{dy} \frac{y^2+x^2}{\sqrt{z^2+y^2+x^2}}, \frac{d}{dz} \frac{y^2+x^2}{\sqrt{z^2+y^2+x^2}}\right]$

Notemos que el resultado anterior es una lista del tipo [x,y,z], que es la manera como el programa maneja los vectores. En este caso, el primer elemento de la lista es la componente x del vector gradiente. De la misma manera para las componentes y y z. Evaluamos las derivadas de la última salida:

→ **ev(%, diff);**
 (%o5) $\left[\frac{\frac{2x}{\sqrt{z^2+y^2+x^2}} - \frac{x(y^2+x^2)}{(z^2+y^2+x^2)^{3/2}}, \frac{2y}{\sqrt{z^2+y^2+x^2}} - \frac{y(y^2+x^2)}{(z^2+y^2+x^2)^{3/2}}, -\frac{(y^2+x^2)z}{(z^2+y^2+x^2)^{3/2}}\right]$

En este punto podemos intentar simplificar las expresiones anteriores, esto lo hacemos con el comando "ratsimp" y así obtener finalmente el vector gradiente:

→ **ratsimp(%);**
 (%o6) $\left[\frac{\sqrt{z^2+y^2+x^2}(2xz^2+xy^2+x^3)}{z^4+(2y^2+2x^2)z^2+y^4+2x^2y^2+x^4}, \frac{\sqrt{z^2+y^2+x^2}(2yz^2+y^3+x^2y)}{z^4+(2y^2+2x^2)z^2+y^4+2x^2y^2+x^4}, -\frac{(y^2+x^2)z}{(z^2+y^2+x^2)^{3/2}}\right]$

Hagamos uso de los otros operadores. Por ejemplo, dado un vector calculemos la divergencia, $\nabla \cdot a$. Esta vez combinaremos algunos comandos en una misma linea.

```

→      a:[x^2/(x^2+y^2),y^2/(x^2+y^2),z^2/(x^2+y^2)];
(a)    [x^2/(y^2+x^2),y^2/(y^2+x^2),z^2/(y^2+x^2)]
→      express(div (a));
(%o8)  d z z^2 + d y y^2 + d x x^2
      ----- + ----- + -----
      y^2 + x^2   y^2 + x^2   y^2 + x^2
→      ev(%, diff),ratsimp;
(%o9)  (2 y^2 + 2 x^2) z + 2 x y^2 + 2 x^2 y
      -----
      y^4 + 2 x^2 y^2 + x^4

```

Calculemos ahora el rotor de \mathbf{a} , $\nabla \times \mathbf{a}$, pero combinado más comandos en una sola instrucción:

```

→      ev(express(curl(a)),diff);
(%o10) [- 2 y z^2 / (y^2 + x^2)^2, - 2 x z^2 / (y^2 + x^2)^2, - 2 x^2 y / (y^2 + x^2)^2 - 2 x y^2 / (y^2 + x^2)^2]

```

Intentemos simplificar nuevamente para ver lo que ocurre:

```

→      ratsimp(%);
(%o11) [- 2 y z^2 / (y^4 + 2 x^2 y^2 + x^4), - 2 x z^2 / (y^4 + 2 x^2 y^2 + x^4), - 2 x y^2 - 2 x^2 y / (y^4 + 2 x^2 y^2 + x^4)]

```

Finalmente, calculemos el laplaciano de f , todo en una sola instrucción

```

→      ratsimp(ev(express(laplacian(f)),diff));
(%o12) 4 z^2 / (z^2 + y^2 + x^2)^{3/2}

```

Podemos verificar rápidamente que el laplaciano $f = \nabla \cdot \nabla f$

```

→      div(grad(f));
(%o13) div grad(y^2 + x^2 / sqrt(z^2 + y^2 + x^2))

```

```

→ ratsimp(ev(express((%)),diff));
(%o14) 
$$\frac{4 z^2}{(z^2 + y^2 + x^2)^{3/2}}$$


→ kill(all)$

```

Maxima integra simbólicamente funciones por medio del comando "integrate(expr,x,a,b)". Esto es, calcula la integral definida de la expr respecto de x con los límites de integración a y b. Si se omiten los límites de integración se calcula la integral indefinida. Por ejemplo

```

→ integrate(a·sin(x+b)^3,x);
(%o1) 
$$a \left( \frac{\cos(x+b)^3}{3} - \cos(x+b) \right)$$


```

Calculemos ahora el área de la región comprendida entre las funciones: $f(x)=x^3-3x^2+1$ y $g(x)=-x+1$. Aprovecharemos este ejercicio para aprender, entre otras cosas, a graficar y definir funciones. Primero que todo, debemos introducir al programa las dos funciones y a diferencia de los cálculos anteriores, usaremos := para definirlas.

```

→ f(x):=x^3-3·x^2+1; g(x):=-x+1;
(%o2) f(x):=x^3 - 3 x^2 + 1
(%o3) g(x):= - x + 1

```

La diferencia de estas dos funciones la podemos llamar la función d(x).

```

→ d(x):=f(x)-g(x);
(%o4) d(x):=f(x)-g(x)


```

Haremos la gráfica de las dos funciones. Maxima permite hacer gráficos en 2D con el comando "plot2d". Pero aquí utilizaremos una variante y haremos el gráfico con el comando "wxplot2d" para que la figura aparezca embebida dentro de nuestra hoja de cálculo.

Es necesario introducir las funciones como una lista, es decir, dentro de corchetes. Por otra parte, le pediremos al programa que grafique para los valores comprendidos de: $-1 \leq x \leq 3$. Posteriormente, necesitaremos encontrar los puntos donde las funciones se interceptan. Para este fin, utilizaremos el comando "solve", que resuelve la ecuación algebraica y devuelve una lista de igualdades con las variables despejadas.

Si la expresión a resolver no es una igualdad, se supone que se quiere resolver la ecuación ya igualada a cero. Entonces, para graficar las funciones $f(x)$ y $g(x)$ entre $[-1,3]$ escribimos el siguiente comando:

→ `wxplot2d([f(x),g(x)],[x,-1,3]);`



Calculemos los puntos donde se interceptan las curvas:

→ `solve(f(x)-g(x),x);`

(%o6) $[x = -\frac{\sqrt{5}-3}{2}, x = \frac{\sqrt{5}+3}{2}, x = 0]$

Los valores numéricos son:

→ `float(%);`

(%o7) $[x = 0.3819660112501051, x = 2.618033988749895, x = 0.0]$

Procedemos ahora sí a integrar la función diferencia para encontrar el área contenida dentro de las dos funciones.

```

→      integrate(d(x),x,0,-(sqrt(5)-3)/2)+integrate(-d(x),x,-(sqrt(5)-3)/2,(sqrt(5)+3)/2)
(%o8)   
$$\frac{5^{3/2} + 11}{8} + \frac{5^{3/2} - 11}{4}$$

→      float(%);
(%o9)  2.817627457812107

```

Calculemos ahora las dos integrales que aparecen en el ejemplo 8 de 1.5.7. Escribiremos primero los dos integrandos:

```

→      int1:(3·x^2+2·x·y^3)·dx; int2:6·x·y·dy;
(int1)  dx (2 x y^3 + 3 x^2)
(int2)  6 dy x y

```

Realizaremos el cambio de variable a través de comando "subst" que nos permite sustituir expresiones dentro de otras:

```

→      e1:subst([x=2·t^2,dx=4·t,y=t^3+t,dy=3·t^2+1],int1);
(e1)   4 t (4 t^2 (t^3 + t)^3 + 12 t^4)

→      e2:subst([x=2·t^2,dx=4·t,y=t^3+t,dy=3·t^2+1],int2);
(e2)   12 t^2 (3 t^2 + 1) (t^3 + t)

```

Podemos introducir las integrales sin evaluarlas en el momento, esto se consigue mediante el operador de comilla simple. En este caso la función no es evaluada y devuelve una expresión simbólica o imagen pictórica.

```

→      int1:'integrate(e1,t,0,sqrt(2)/2);
(int1)  
$$4 \int_0^{\frac{1}{\sqrt{2}}} t (4 t^2 (t^3 + t)^3 + 12 t^4) dt$$


```

Ahora si evaluamos la integral:

→ `ev(int1,integrate),expand;`

$$(\%o15) \quad \frac{9305}{3003 2^{5/2}} + 1$$

Y para la segunda integral:

→ `int2:'integrate(e2,t,0,sqrt(2)/2);`

$$(\text{int2}) \quad 12 \int_0^{\frac{1}{\sqrt{2}}} t^2 (3t^2 + 1)(t^3 + t) dt$$

→ `ev(int2,integrate),expand;`

$$(\%o17) \quad \frac{65}{32}$$

→ `kill(all)$`

Para finalizar, aclaremos la diferencia entre asignar una expresión a una variable y la definición de funciones. Consideremos el movimiento de una partícula que sigue la trayectoria:

→ `x:x0+v0*t+a*t^2/2;`

$$(x) \quad x0 + t v0 + \frac{a t^2}{2}$$

Si queremos evaluarla para algún valor de la variable t, digamos t = 1, entonces podemos escribir:

→ `subst(t=1,x);`

$$(\%o2) \quad x0 + v0 + \frac{a}{2}$$

Si queremos evaluar la expresión con el resto de parámetros, entonces:

→ `subst([t=1,a=-9.8,x0=0,v0=10],x);`

$$(\%o3) \quad 5.1$$

Las sustituciones se realizan pero no se asignan a las variables. Podemos ver que x sigue siendo lo que le asignamos originalmente.

→ $x;$
 (%o4) $x0 + t v0 + \frac{a t^2}{2}$

Ahora definamos la expresión anterior pero como la función $x(t)$.

→ $x(t) := x0 + v0 \cdot t + a \cdot t^2 / 2;$
 (%o5) $x(t) := x0 + v0 t + \frac{a t^2}{2}$

Evaluarla es ahora más sencillo:

→ $x(1);$
 (%o6) $x0 + v0 + \frac{a}{2}$

También podemos escribir:

→ $\text{subst}([a = -9.8, x0 = 0, v0 = 10], x(1));$
 (%o7) 5.1

Si queremos asignarle los valores a los parámetros de manera global, entonces escribimos:

→ $a: -9.8 \$ x0: 0 \$ v0: 10 \$$

De manera que:

→ $x(1);$
 (%o11) 5.1

Para limpiar los parámetros que acabamos de asignar hacemos lo siguiente:

→ $\text{kill}(a, v0, x0) \$$

Por lo tanto:

→ $x(1);$
 (%o13) $x0 + v0 + \frac{a}{2}$

La velocidad y la aceleración son funciones fáciles de calcular:

$$\rightarrow \text{'diff}(x(t),t)=\text{diff}(x(t),t);$$

(%o14) $\frac{d}{dt} \left(x_0 + t v_0 + \frac{a t^2}{2} \right) = v_0 + a t$

$$\rightarrow \text{'diff}(x(t),t,2)=\text{diff}(x(t),t,2);$$

(%o15) $\frac{d^2}{dt^2} \left(x_0 + t v_0 + \frac{a t^2}{2} \right) = a$

En el caso de que estemos interesados en definir una nueva función a partir de algún cálculo anterior podemos utilizar la función "define". Entonces, lo que vamos a hacer en las siguientes líneas de comandos, es definir la función velocidad a partir de la derivada de la posición.

$$\begin{aligned} \rightarrow & \quad \text{diff}(x(t),t); \\ (%o16) & \quad v_0 + a t \\ \rightarrow & \quad \text{define}(v(t), \%); \\ (%o17) & \quad v(t) := v_0 + a t \\ \rightarrow & \quad v(1); \\ (%o18) & \quad v_0 + a \\ \rightarrow & \quad \text{subst}([a=-9.8, x_0=0, v_0=10], v(1)); \\ (%o19) & \quad 0.1999999999999993 \\ \rightarrow & \quad \text{kill}(all)\$ \end{aligned}$$

Cuando necesitemos manejar campos vectoriales podemos utilizar las listas.

Veamos el primer ejemplo de 1.5.7 donde: $r=3t^2i+(4t^3-t)j+tk$. Escribamos el vector posición:

$$\begin{aligned} \rightarrow & \quad [x,y,z]: [3 \cdot t^2, 4 \cdot t^3 - t, t]; \\ (%o1) & \quad [3 t^2, 4 t^3 - t, t] \end{aligned}$$

La velocidad:

$$\begin{aligned} \rightarrow & \quad v: \text{diff}([x,y,z],t); \\ (%o) & \quad [6 t, 12 t^2 - 1, 1] \end{aligned}$$

La aceleración:

```
→ a: diff([x,y,z],t,2);
(a) [ 6, 24 t, 0 ]
```

Podemos facilmente calcular:

```
→ v.[bx,by,bz];
(%o4) by (12 t2 - 1) + 6 bx t + bz
→ solve(% ,bz);
(%o5) [ bz = -12 by t2 - 6 bx t + by ]
→ kill(all)$
```

FIN

6 - 1.6.6 Vectores y números complejos

Maxima maneja números complejos escritos en la forma: $a + bi$. La unidad imaginaria es interpretada por el programa como $i = \%i$. Si no queremos que las salidas del programa mantenga la notación del $\%i$ podemos ir a la parte superior de la ventana del programa y hacer click en:

wxMaxima > Preferencias > Worksheet y desactivar el botón que dice

[] Mantener signo porcentual en símbolos especiales: $\%e$, $\%i$, etc.

1. Algunos cálculos básicos.

Los números imaginarios aparecen si queremos calcular la raíz cuadrada de un número negativo, por ejemplo, ejemplo, la raíz cuadrada de -7

```
→ sqrt(-7);
(%o1) √7 i
```

El programa nos permite desarrollar toda el álgebra en variable compleja. Si queremos sumar $z1 = 1+2i$ y $z2 = 3 + 4i$, escribimos:

```

→      z1:1+2·%i; z2:3+4·%i;
(z1)   2 i + 1
(z2)   4 i + 3

→      z1+z2;
(%o4) 6 i + 4

```

La multiplicación:

```

→      z1·z2,expand;
(%o5) 10 i - 5

```

Y la división:

```

→      z1/z2;
(%o6) 2 i + 1
                  -----
                  4 i + 3

→      rectform(%);
(%o7) 2 i + 11
                  -
                  25

```

Este último comando pertenece a una lista de funciones de Maxima para el cálculo de números complejos:

rectform(expresión)	expresión en forma cartesiana o binómica
realpart(expresión)	parte real de expresión
imagpart(expresión)	parte imaginaria de expresión
polarform(expresión)	forma polar de expresión
abs(expresión)	módulo o valor absoluto de expresión
cabs(expresión)	módulo de expresión compleja
carg(expresión)	argumento de expresión
conjugate(expresión)	conjuguado de expresión
demoivre(expresión)	expresa el número complejo utilizando senos y cosenos
exponentialize(expresión)	expresa el número complejo utilizando exponentiales

Veamos como funcionan algunas de estos comandos:

```

→      realpart(z1); imagpart(z1);
(%o8)  1
(%o9)  2

→      abs(z1);
(%o10)  $\sqrt{5}$ 

→      polarform(z1);
(%o11)  $\sqrt{5} e^{i \operatorname{atan}(2)}$ 

→      cabs(z1);carg(z1);
(%o12)  $\sqrt{5}$ 
(%o13)  $\operatorname{atan}(2)$ 

→      conjugate(z1);
(%o14)  $1 - 2i$ 

```

2. Para una función, digamos $\tan(x + iy)$, la podemos escribir como exponenciales o funciones senos y cosenos.

```

→      tan(x+%i·y);
(%o15)  $\tan(i y+x)$ 

→      exponentialize(%),factor;
(%o16) 
$$\frac{i (e^y - e^{i x}) (e^y + e^{i x})}{e^{2 y} + e^{2 i x}}$$


→      demoivre(%),factor;
(%o17) 
$$\frac{i (e^y - i \sin(x) - \cos(x)) (e^y + i \sin(x) + \cos(x))}{e^{2 y} + i \sin(2 x) + \cos(2 x)}$$


→      log(3+%i·2);
(%o18)  $\log(2i+3)$ 

→      cabs(%);
(%o19) 
$$\sqrt{\frac{\log(13)^2}{4} + \operatorname{atan}\left(\frac{2}{3}\right)^2}$$


```

```
→      float(%);
(%o20) 1.410846683153171
```

3. Tomemos dos números complejos, o vectores del plano complejo

```
→      z:a+b·%i; w:c+d·%i;
(z)    i b+a
(w)    i d+c
```

Revisemos la llamada igualdad del paralelogramo: la suma de los cuadrados de las longitudes de los cuatro lados de un paralelogramo, es igual a la suma de los cuadrados de las longitudes de las dos diagonales de éste. Es decir:

$$|z + w|^2 + |z - w|^2 = 2 |z|^2 + |w|^2$$

Pero sabemos que $|z|^2 = zz^*$, y por lo tanto:

$$(w+z)(w+z)^* + (w-z)(w-z)^* \\ = 2|z|^2 + |w|^2$$

Calculemos el lado izquierdo de la ecuación anterior:

```
→      (w+z)·conjugate(w+z)+(w-z)·conjugate(w-z),expand;
(%o23) 2 d^2+2 c^2+2 b^2+2 a^2
```

Ahora el lado derecho, es decir, la suma de las longitudes al cuadrado de los cuatro lados:

```
→      2·cabs(z)^2+2·cabs(w)^2,expand;
(%o24) 2 d^2+2 c^2+2 b^2+2 a^2
```

4. Para encontrar las raíces de un número complejo, debemos declarar a la variable como compleja.

```
→      kill(z)$
→      declare(z,complex);
(%o26) done
```

De manera que si queremos encontrar las raíces de $z^3 = 1$

```
→      ec1:z^3=1;
(ec1)   z^3=1

→      solve(ec1,z);
(%o28)  [z=-(sqrt(3)*i - 1)/2, z=-sqrt(3)*i + 1/2, z=1]
```

5. En otro de los ejemplo que discutimos anteriormente teníamos la ecuación:

```
→      ec2:z^6-z^5+4*z^4-6*z^3+2*z^2-8*z+8=0;
(ec2)   z^6 - z^5 + 4 z^4 - 6 z^3 + 2 z^2 - 8 z + 8 = 0

→      factor(ec2);
(%o30)  (z - 1) (z^2 + 4) (z^3 - 2) = 0

→      solve(ec2,z),factor;
(%o31)  [z=-(sqrt(3)*i - 1)/2^(2/3), z=-sqrt(3)*i + 1/2^(2/3), z=2^(1/3), z=1, z=-2*i, z=2*i]

→      kill(all)$
```

FIN