

1.1 Espacios vectoriales lineales

7/11/2019/

1 - 2.1.7 Grupos,campos y espacios vectoriales

En este módulo utilizaremos algunas de las herramientas disponibles para incorporar el álgebra discreta. Comenzaremos con la introducción a los conjuntos. Podemos operar con conjuntos pero primero debemos definirlos. Existen varias posibilidades, como mostramos a continuación:

- A:{1,2,3,4,5,6,7,8,9};
- (A) {1,2,3,4,5,6,7,8,9}

- B:set(1,3,5,7,9);
- (B) {1,3,5,7,9}

- C:set(2,4,6,8);
- (C) {2,4,6,8}

- D:makeset(i/j, [i,j], [[1,3], [2,3], [3,3], [4,3]]);
- (D) $\{\frac{1}{3}, \frac{2}{3}, 1, \frac{4}{3}\}$

Notemos que es igual definir los conjuntos con llaves, con la función set o makeset. Podemos preguntar si determinado elemento pertenece, o no, a un conjunto.

- elementp(7,A);
- (%o5) true

- elementp(7,C);
- (%o6) false

Operaciones elementales con conjuntos

- UBC:union(B,C);
- (UBC) {1,2,3,4,5,6,7,8,9}

```
→      is (A=UBC);  
(%o8)  true  
  
→      Cv:intersection(B,C);  
(Cv)  {}
```

Para Maxima el conjunto vacío es {}.

```
→      setdifference(A,C);  
(%o10) {1,3,5,7,9}
```

Esto es, el conjunto con los elementos del conjunto A que no pertenecen al conjunto C.

Si queremos el conjunto de todos los subconjuntos del conjunto {a, b, c} le podemos pedir al programa que nos lo muestre:

```
→      powerset({a,b,c});  
(%o11) {{}, {a}, {a, b}, {a, b, c}, {a, c}, {b}, {b, c}, {c}}
```

El producto cartesiano de los conjuntos A y B es el conjunto conformado por los pares (a, b):

$$A \times B = \{(a, b) / a \in A, b \in B\}.$$

```
→      cartesian_product(B,C);  
(%o12) {[1,2],[1,4],[1,6],[1,8],[3,2],[3,4],[3,6],[3,8],[5,2]  
],[5,4],[5,6],[5,8],[7,2],[7,4],[7,6],[7,8],[9,2],[9,4],[9,6],[  
9,8]}
```

Le podemos pedir al programa la suma de los pares del producto cartesiano anterior:

```
→      makeset(a+b, [a,b], cartesian_product(B,C));  
(%o13) {3,5,7,9,11,13,15,17}
```

Maxima trata a los conjuntos y a las listas como objetos de distinta naturaleza, lo que permite trabajar con conjuntos cuyos elementos puedan ser también conjuntos o listas, es decir, subconjuntos.

```
→      lista:makelist(i,i,1,30);  
(lista) [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20  
,21,22,23,24,25,26,27,28,29,30]
```

La lista anterior la convertiremos en un conjunto, para este fin debemos utilizar el comando "setify"

```
→      E:setify(lista);
(E)    {1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,
20,21,22,23,24,25,26,27,28,29,30}
```

De este último conjunto podemos construir el subconjunto conformado por los números primos:

```
→      Primos:subset(E,primep);
(Primos) {2,3,5,7,11,13,17,19,23,29}
```

Con "cardinality" podemos saber cuantos elementos contiene un conjunto:

```
→      cardinality(%);
(%o17) 10
```

La función de Maxima que nos permite calcular las tablas que utilizamos en la sección 2.1 es la función "mod". Veamos:

```
→      Gm8:{1,3,5,7};
(Gm8) {1,3,5,7}
```

Todos los productos posibles entre los elementos de este conjunto son:

```
→      makeset(a.b, [a,b], cartesian_product(Gm8,Gm8));
(%o19) {1,3,5,7,9,15,21,25,35,49}
```

Los modulo 8 para algunos de los números anteriores son:

```
→      mod(21,8);mod(35,8);mod(49,8);
(%o20) 5
(%o21) 3
(%o22) 1
```

Para generar el grupo G[mod5] escribimos:

```
→      setify(makelist(mod(i,5),i,1,4));
(%o23) {1,2,3,4}
```

En la sección 2.1.1, definimos el orden de un elemento $g \in G$. Consideremos el siguiente conjunto de números enteros:

$$Z_{15} = \{0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14\}.$$

Se podría demostrar que el orden de $g \in G$ es igual al número de elementos de $\langle g \rangle$ y lo dejaremos como ejercicio.

Vamos a calcular el orden de todos los elementos de Z_{15} , sabiendo que el orden de cada uno de esos elementos divide a 15, que es el cardinal de Z_{15} .

Probemos primero con el número $6 \in Z_{15}$.

```
→      setify(makelist(mod(6·i,15),i,0,14));
(%o24)  {0, 3, 6, 9, 12}
```

El orden de $6 \in Z_{15}$ es:

```
→      length(%);
(%o25)  5
```

En la siguiente instrucción lo haremos para todos los elementos de Z_{15} .

```
→      makelist([j,length(setify(makelist(mod(j·i,15),i,0,14))))],j,0,14);
(%o26)  [[0,1],[1,15],[2,15],[3,5],[4,15],[5,3],[6,5],[7,15],[8,15],[9,5],[10,3],[11,15],[12,5],[13,15],[14,15]]
```

La salida no es más que una lista $[x, y]$ con cada elemento $x \in Z_{15}$ y su orden y . Por lo tanto, el conjunto de órdenes es: $\{1, 3, 5, 15\}$, todos divisores de 15, como estipula el teorema de Lagrange.

```
→      kill(all)$
```

FIN

2 - 2.2.5 Espacios métricos, normados y con producto interno

1- Espacios y subespacios vectoriales:

Sea el espacio vectorial $V = K^n$, definido en $K = \mathbb{R}$ y donde n es un entero positivo. Consideremos el caso $n = 4$. El producto de un elemento de K^4 , digamos $|x\rangle = (x_1, x_2, x_3, x_4)$ por un escalar $\alpha \in K$ resulta en otro elemento de K^4 .

Primero introducimos los elementos como listas:

```

→      X:[x1,x2,x3,x4];Y:[y1,y2,y3,y4];Z:[z1,z2,z3,z4];
(X)    [x1,x2,x3,x4]
(Y)    [y1,y2,y3,y4]
(Z)    [z1,z2,z3,z4]

→      α·X=Y;
(%o4)  [x1 α, x2 α, x3 α, x4 α]=[y1, y2, y3, y4]
```

El resultado es un elemento del espacio vectorial K^4 . La suma de $|x\rangle = (x_1, x_2, x_3, x_4)$ y $|y\rangle = (y_1, y_2, y_3, y_4)$ será:

```

→      X+Y=Z;
(%o5)  [y1+x1, y2+x2, y3+x3, y4+x4]=[z1, z2, z3, z4]
```

con $(z_1, z_2, z_3, z_4) \in K^4$. Podemos ver rápidamente que el conjunto de vectores que tienen la forma $(x_1, x_2, x_3, 0)$ conforman un subespacio de K^4 , ya que:

```

→      map(":",[x4,y4,z4],[0,0,0]);
(%o6)  [0,0,0]

→      X:[x1,x2,x3,x4];Y:[y1,y2,y3,y4];Z:[z1,z2,z3,z4];
(X)    [x1,x2,x3,0]
(Y)    [y1,y2,y3,0]
(Z)    [z1,z2,z3,0]

→      α·X+β·Y=Z;
(%o10)  [y1 β+x1 α, y2 β+x2 α, y3 β+x3 α, 0]=[z1, z2, z3, 0]
```

Para recobrar las variables x_4, y_4, z_4 escribimos:

```
→      kill(x4,y4,z4)$
```

```

→      X:[x1,x2,x3,x4];Y:[y1,y2,y3,y4];Z:[z1,z2,z3,z4];
(X)    [x1,x2,x3,x4]
(Y)    [y1,y2,y3,y4]
(Z)    [z1,z2,z3,z4]

```

Para calcular el producto interno entre vectores es necesario utilizar la librería "eigen".

```

→      load("eigen")$

→      innerproduct(X,Y);
(%o16) x4 y4+x3 y3+x2 y2+x1 y1

```

Consideremos ahora $V = K^n$, definido en $K = C$, con $n = 3$. Por lo tanto, los vectores serán ahora de la siguiente forma:

$$z = (x_1 + iy_1, x_2 + iy_2, x_3 + iy_3).$$

```

→      Z1:[x1+%i·y1,x2+%i·y2,x3+%i·y3]; Z2:[u1+%i·v1,u2+%i·v2,u3+%i·v3];
(Z1)   [i y1+x1,i y2+x2,i y3+x3]
(Z2)   [i v1+u1,i v2+u2,i v3+u3]

```

Y los escalares de la forma $\alpha = a + ib$.

```

→      α:a+%i·b;
(α)   i b+a

```

El producto por el escalar α es:

```

→      Z3: α·Z1;
(Z3)   [(i b+a)(i y1+x1),(i b+a)(i y2+x2),(i b+a)(i y3+x3)]

→      map(rectform,Z3);
(%o21) [i(a y1+b x1)-b y1+a x1,i(a y2+b x2)-b y2+a x2,i
(a y3+b x3)-b y3+a x3]

→      map(realpart,Z3),factor; map(imagpart,Z3),factor;
(%o22) [-(b y1-a x1),-(b y2-a x2),-(b y3-a x3)]
(%o23) [a y1+b x1,a y2+b x2,a y3+b x3]

```

Calculemos ahora el producto interno:

$$Z1 \cdot Z2 = (x_1 + iy_1) * (u_1 + iv_1) + (x_2 + iy_2) * (u_2 + iv_2) + (x_3 + iy_3) * (u_3 + iv_3)$$

→ **Z4:innerproduct(Z1,Z2);**

$$(Z4) (v_3 - i u_3) y_3 + (v_2 - i u_2) y_2 + (v_1 - i u_1) y_1 + (i v_3 + u_3) x_3 + (i v_2 + u_2) x_2 + (i v_1 + u_1) x_1$$

→ **Re:map(realpart,Z4)\$ Im:map(imagpart,Z4)\$**

→ **Re+%i·Im;**

$$(\%o27) i (- u_3 y_3 - u_2 y_2 - u_1 y_1 + v_3 x_3 + v_2 x_2 + v_1 x_1) + v_3 y_3 + v_2 y_2 + v_1 y_1 + u_3 x_3 + u_2 x_2 + u_1 x_1$$

→ **kill(all)\$**

2- Producto de polinomios

Consideremos el producto escalar entre elementos de un espacio vectorial de polinomios.

Vamos a encontrar la distancia y el ángulo entre los vectores

$$|x_1\rangle = x(x - 1) \text{ y } |x_2\rangle = x$$

en dos intervalos diferentes: $[0, 1]$ y $[-1, 1]$.

Debemos introducir los objetos a multiplicar

→ **P1:x·(x-1); P2:x;**

$$(P1) (x - 1) x$$

$$(P2) x$$

Ahora calculamos las distancias entre los vectores para ambos intervalos.

Haremos gala de algunas posibilidades que ofrece el programa para escribir las expresiones.

→ **sqrt('integrate(((P1-P2)^2),x,-1,1))=sqrt(integrate(((P1-P2)^2),x,-1,1));**

$$(\%o3) \sqrt{\int_{-1}^1 ((x - 1) x - x)^2 dx} = \frac{\sqrt{46}}{\sqrt{15}}$$

$$\rightarrow \text{sqrt('integrate(((P1-P2)^2),x,0,1))=sqrt(integrate(((P1-P2)^2),x,0,1));}$$

$$(\%o4) \quad \sqrt{\int_0^1 ((x-1)x-x)^2 dx} = \frac{2^{3/2}}{\sqrt{15}}$$

El ángulo entre los polinomios definidos en el intervalo $[-1, 1]$ es:

$$\rightarrow \text{'integrate((P1·P2),x,-1,1)/(sqrt('integrate((P1·P1),x,-1,1))·sqrt('integrate((P2·P2),x,-1,1)))}$$

$$(\%o5) \quad \frac{\int_{-1}^1 (x-1)x^2 dx}{\sqrt{\int_{-1}^1 x^2 dx} \sqrt{\int_{-1}^1 (x-1)^2 x^2 dx}}$$

$$\rightarrow \text{ev(%integrate);}$$

$$(\%o6) \quad -\frac{\sqrt{15}}{2^{3/2} \sqrt{3}}$$

$$\rightarrow \text{acos(%),numer;}$$

$$(\%o7) \quad 2.482534617763384$$

Y ahora, el ángulo entre los polinomios definidos en el intervalo $[0, 1]$:

$$\rightarrow \text{'integrate((P1·P2),x,0,1)/(sqrt('integrate((P1·P1),x,0,1))·sqrt('integrate((P2·P2),x,0,1)))}$$

$$(\%o8) \quad \frac{\int_0^1 (x-1)x^2 dx}{\sqrt{\int_0^1 x^2 dx} \sqrt{\int_0^1 (x-1)^2 x^2 dx}}$$

$$\rightarrow \text{ev(%integrate);}$$

$$(\%o9) \quad -\frac{\sqrt{30}}{4 \sqrt{3}}$$

$$\rightarrow \text{acos(%),numer;}$$

$$(\%o10) \quad 2.482534617763385$$

$$\rightarrow \text{kill(all)$}$$

FIN

3 - 2.3.7 Variedades lineales

1.- Independencia Lineal

En 2.3.1 vimos que si en la ecuación $|0\rangle = C_1 |v_1\rangle + C_2 |v_2\rangle + C_3 |v_3\rangle + \dots + C_n |v_n\rangle$ con todos los $C_i = 0$, entonces se dirá que el conjunto de vectores son linealmente independientes. Para el primer ejemplo de esa sección se obtuvo el siguiente sistema de ecuaciones:

$$\begin{aligned} C_1 + 2C_2 - C_3 &= 0 \\ 3C_1 + C_3 &= 0 \\ -C_1 + C_2 &= 0 \\ 2C_1 + 3C_2 &= 0 \end{aligned}$$

que procedemos a resolver usando el comando "linsolve":

```
→      linsolve([C1+2·C2-C3=0, 3·C1+C3=0, -C1+C2=0, 2·C1+3·C2=0], [C1,C2,C3,C4])
solve: dependent equations eliminated: (4)
(%o1)      [C1=0, C2=0, C3=0, C4=%r1]
```

2.- Bases para espacios vectoriales

En este ejercicio aprenderemos a calcular una base a partir de un conjunto de vectores perteneciente a un determinado espacio vectorial. Por ejemplo, si en \mathbb{R}^5 tenemos el siguiente conjunto de vectores:

$$|v_1\rangle = (1, 2, 3, 4, 5), |v_2\rangle = (0, -1, 1, 2, 3), |v_3\rangle = (3, 2, 1, 0, -1), |v_4\rangle = (-4, -3, -2, -1, 0).$$

```
→      v1:[1,2,3,4,5]; v2:[0,-1,1,2,3]; v3:[3,2,1,0,-1]; v4:[-4,-3,-2,-1,0];
(v1)      [1,2,3,4,5]
(v2)      [0,-1,1,2,3]
(v3)      [3,2,1,0,-1]
(v4)      [-4,-3,-2,-1,0]
```

Con los vectores dados construimos la siguiente matriz

→ `M:matrix(v1,v2,v3,v4);`

(M)

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 0 & -1 & 1 & 2 & 3 \\ 3 & 2 & 1 & 0 & -1 \\ -4 & -3 & -2 & -1 & 0 \end{bmatrix}$$

Como veremos más adelante, el Rango de una matriz indica el número máximo de vectores fila o columna linealmente independientes

→ `rank(M);`

(%o7) 3

Podemos aplicar el método de eliminación gaussiana a la matriz M para obtener una nueva matriz escalonada. El cálculo además se hace normalizando el primer elemento no nulo de cada fila. Esto se logra con el comando "echelon".

→ `echelon(M);`

(%o8)

$$\begin{bmatrix} 1 & \frac{2}{3} & \frac{1}{3} & 0 & -\frac{1}{3} \\ 0 & 1 & -1 & -2 & -3 \\ 0 & 0 & 1 & \frac{5}{3} & \frac{7}{3} \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Por lo tanto, cada fila de la matriz anterior conformará un conjunto de vectores linealmente independiente. Los podemos aislar de la matriz con el comando "row(M,i)", que devolverá la i-ésima fila de la matriz M.

→ `e1:row(%o8,1);`

(e1)

$$\begin{bmatrix} 1 & \frac{2}{3} & \frac{1}{3} & 0 & -\frac{1}{3} \end{bmatrix}$$

→ `e2:row(%o8,2);`

(e2)

$$\begin{bmatrix} 0 & 1 & -1 & -2 & -3 \end{bmatrix}$$

→ `e3:row(%o8,3);`

(e3)

$$\begin{bmatrix} 0 & 0 & 1 & \frac{5}{3} & \frac{7}{3} \end{bmatrix}$$

Note que usamos $\%o8$ para usar la salida producida en ($\%i8$)
 Es trivial verificar que el conjunto: $\{e_1, e_2, e_3\}$ es linealmente independientes.

→ $\text{solve}(\alpha \cdot [1, 2/3, 1/3, 0, -1/3] + \beta \cdot [0, 1, -1, -2, -3] + \gamma \cdot [0, 0, 0, 5/3, 7/3], [\alpha, \beta, \gamma]);$
 solve: dependent equations eliminated: (3 4)
 $(\%o12) [[\alpha=0, \beta=0, \gamma=0]]$

Consideremos los vectores $a = (1, 3)$ y $b = (-1, 1)$ ¿Serán linealmente independientes?

→ $\text{solve}(\alpha \cdot [1, 3] + \beta \cdot [-1, 1], [\alpha, \beta]);$
 $(\%o13) [[\alpha=0, \beta=0]]$

Los vectores $a = (1, 2, 3)$ y $b = (4, 8, 12)$ ¿Serán linealmente independientes?

→ $\text{solve}(\alpha \cdot [1, 2, 3] + \beta \cdot [4, 8, 12], [\alpha, \beta]);$
 solve: dependent equations eliminated: (2 3)
 $(\%o14) [[\alpha=-4 \%r2, \beta=%r2]]$

Aquí $\alpha = -4\beta$.

Sea ahora $\{e_i\} = \{(1, 1, 1), (1, 2, 1), (0, 0, 2)\}$ una base para \mathbb{R}^3 . Vamos a calcular las componentes del vector $a = (3, 2, 1)$ respecto de esa base.

Primero verifiquemos si efectivamente es una base:

→ $\text{solve}(\alpha \cdot [1, 1, 1] + \beta \cdot [1, 2, 1] + \gamma \cdot [0, 0, 2], [\alpha, \beta, \gamma]);$
 $(\%o15) [[\alpha=0, \beta=0, \gamma=0]]$

→ $\text{solve}([3, 2, 1] - ax \cdot [1, 1, 1] - ay \cdot [1, 2, 1] - az \cdot [0, 0, 2], [ax, ay, az]);$
 $(\%o16) [[ax=4, ay=-1, az=-1]]$

Por lo tanto, $a = 4e_1 - e_2 - e_3$.

En la base canónica las componentes del vector a serán:

→ $\text{solve}([3, 2, 1] - ax \cdot [1, 0, 0] - ay \cdot [0, 1, 0] - az \cdot [0, 0, 1], [ax, ay, az]);$
 $(\%o17) [[ax=3, ay=2, az=1]]$

Es decir, $a = 3i + 2j + k$.

→ kill(all)\$

3.- Ortogonalización con Maxima

Anteriormente hicimos los cálculos para hallar una base ortogonal a partir de un conjunto de vectores linealmente independientes.

$$|v_1\rangle = (1, 3, -1, 2), |v_2\rangle = (2, 0, 1, 3), |v_3\rangle = (-1, 1, 0, 0)$$

La función "gramschmidt(x)" es la que ejecuta el algoritmo de ortogonalización de Gram-Schmidt sobre el objeto x. Si x es una matriz el algoritmo actúa sobre sus filas. Antes de usar la función es necesario cargar la librería "eigen"

→ load("eigen")\$

→ v:matrix ([-1, 1, 0, 0], [2, 0, 1, 3], [1, 3, -1, 2]);
(v)
$$\begin{bmatrix} -1 & 1 & 0 & 0 \\ 2 & 0 & 1 & 3 \\ 1 & 3 & -1 & 2 \end{bmatrix}$$

Notemos que el vector $|v_3\rangle$ es el que hemos puesto en la primera fila de la matriz. Ahora procedemos al cálculo:

→ e:gramschmidt(v);

(e) $[[-1, 1, 0, 0], [1, 1, 1, 3], [\frac{5}{2^2}, \frac{5}{2^2}, -\frac{7}{2^2}, -\frac{1}{2^2}]]$

El resultado viene en forma de una lista: [e[1], e[2], e[3]]. Simplificando resulta:

→ e:expand(%);

(e) $[[-1, 1, 0, 0], [1, 1, 1, 3], [\frac{5}{4}, \frac{5}{4}, -\frac{7}{4}, -\frac{1}{4}]]$

Podemos verificar que son ortogonales:

→ map(innerproduct, [e[1], e[2], e[3]], [e[2], e[3], e[1]]);

(%o5) [0, 0, 0]

Normalizamos ahora cada uno de los vectores

```

→      e[1]/sqrt(innerproduct(e[1],e[1]));
(%o6)  [-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0, 0]

→      e[2]/sqrt(innerproduct(e[2],e[2]));
(%o7)  [\frac{1}{2 \sqrt{3}}, \frac{1}{2 \sqrt{3}}, \frac{1}{2 \sqrt{3}}, \frac{\sqrt{3}}{2}]

→      e[3]/sqrt(innerproduct(e[3],e[3]));
(%o8)  [\frac{1}{2}, \frac{1}{2}, -\frac{7}{10}, -\frac{1}{10}]

```

La función "gramschmidt(x,F)" nos permite definir un producto interno diferente a "innerproduct". Veamos como se hace para el otro ejemplo donde el conjunto de vectores linealmente independientes estaba dado por:
 $\{1, t, t^2, t^3, \dots, t^n\}$.

```

→      producto(f,g):=integrate(f·g, t, a, b);
(%o9)  producto(f, g):=\int_a^b f g dt

→      e:gramschmidt ([1, t, t^2, t^3, t^4], producto), a= -1, b=1;
(e)   [1, t, \frac{3 t^2 - 1}{3}, \frac{t (5 t^2 - 3)}{5}, \frac{35 t^4 - 30 t^2 + 3}{35}]

→      e:expand(e);
(e)   [1, t, t^2 - \frac{1}{3}, t^3 - \frac{3 t}{5}, t^4 - \frac{6 t^2}{7} + \frac{3}{35}]

```

Verifiquemos si son ortogonales:

```

→      map(producto, [e[1], e[2], e[3]], [e[2], e[3], e[1]]), a= -1, b=1;
(%o12) [0, 0, 0]

```

Para normalizar hacemos lo siguiente:

```

→      a: -1$ b:1$

```

```

→      e[1]/sqrt(producto(e[1], e[1]));
(%o15)   $\frac{1}{\sqrt{2}}$ 

→      e[2]/sqrt(producto(e[2], e[2]));
(%o16)   $\frac{\sqrt{3} t}{\sqrt{2}}$ 

→      e[3]/sqrt(producto(e[3], e[3])),expand;
(%o17)   $\frac{3 \sqrt{5} t^2}{2^{3/2}} - \frac{\sqrt{5}}{2^{3/2}}$ 

→      e[4]/sqrt(producto(e[4], e[4])),expand;
(%o18)   $\frac{5 \sqrt{7} t^3}{2^{3/2}} - \frac{3 \sqrt{7} t}{2^{3/2}}$ 

→      kill(all)$

```

FIN

4 - 2.4.6 Aproximación de funciones

1.- Series de Fourier

Existe una librería llamada "fourie" en Maxima que contiene instrucciones para el cálculo simbólico de series de Fourier de una función $f(x)$ en el intervalo $[-l,l]$: "fourier (f,x,l)".

También encontraremos un conjunto de comandos en el paquete para calcular los coeficientes y para manipular las expresiones resultantes.

```
→      load(fourie)$
```

En el ejemplo anterior aproximamos la función: $f(x) = x^2$.

Veamos como se trabaja con el programa para calcular la serie de Fourier. Los resultados aparecerán en la forma de listas temporales y entre ellas los coeficientes. Las listas temporales serán indicadas con la notación (%t).

```

→ f:x^2;
(f) x^2

→ fourier(f,x,%pi);
(%t3) a_0 =  $\frac{\pi^2}{3}$ 
(%t4) a_n =  $\frac{2 \left( \frac{\pi^2 \sin(\pi n)}{n} - \frac{2 \sin(\pi n)}{n^3} + \frac{2 \pi \cos(\pi n)}{n^2} \right)}{\pi}$ 
(%t5) b_n = 0
(%o5) [ %t3, %t4, %t5]

```

Lo anterior se puede simplificar con el comando "foursimp":

```

→ foursimp(%);
(%t6) a_0 =  $\frac{\pi^2}{3}$ 
(%t7) a_n =  $\frac{4 (-1)^n}{n^2}$ 
(%t8) b_n = 0
(%o8) [ %t6, %t7, %t8]

```

Podemos evaluar la lista de los coeficiente hasta el término k. Aquí lo haremos hasta k = 4 y el resultado lo asignaremos a la variable F. Por otro lado, usaremos (%o8), la última salida, como entrada para siguiente comando.

```


→ F:fourexpand(%o8,x,%pi,4);
(F)  $\frac{\cos(4x)}{4} - \frac{4 \cos(3x)}{9} + \cos(2x) - 4 \cos(x) + \frac{\pi^2}{3}$ 

```

Construiremos ahora en un mismo gráfico la función original y los primeros 5 términos de la serie, de esta manera podremos comparar el resultado de la aproximación. Las opciones para realizar los diferentes gráficos en Maxima se pueden consultar en el manual de programa.

→ `wxplot2d([F,f], [x,-%pi,%pi],[legend, "F", "x^2"])$`

(%t10)



Veamos que sucede si escribimos:

→ `totalfourier(f,x,%pi);`

$$(%t11) \quad a_0 = \frac{\pi^2}{3}$$

$$(%t12) \quad a_n = \frac{2 \left(\frac{\pi^2 \sin(\pi n)}{n} - \frac{2 \sin(\pi n)}{n^3} + \frac{2 \pi \cos(\pi n)}{n^2} \right)}{\pi}$$

$$(%t13) \quad b_n = 0$$

$$(%t14) \quad a_0 = \frac{\pi^2}{3}$$

$$(%t15) \quad a_n = \frac{4 (-1)^n}{n^2}$$

$$(%t16) \quad b_n = 0$$

$$(%o16) \quad 4 \left(\sum_{n=1}^{\infty} \frac{(-1)^n \cos(n x)}{n^2} \right) + \frac{\pi^2}{3}$$

En este caso fueron aplicados de manera simultánea los comandos "fourier" y "foursimp" para finalmente presentar la serie en forma de una sumatoria.

→ `kill(all)$`

2.- Mínimos Cuadrados

Maxima puede estimar los parámetros que mejor se ajusten a una función $f=(x,y)$ para un conjunto de datos, utilizando el método de mínimos cuadrados. El programa buscará primero una solución exacta, si no la encuentra buscará una aproximada. El resultado lo presentará como una lista de ecuaciones. La función a utilizar será "lsquares".

Vamos a considerar los ejemplos estudiados con anterioridad:

1. En el primer ejemplo los datos eran los siguientes:

$$(x,y) = (1,2),(3,2),(4,5),(6,6)$$

Es necesario hacer uso de la librería "lsquares" y los los datos deben introducirse en forma de matriz

```
→ load(lsquares)$
→ datos: matrix([1,2],[3,2],[4,5],[6,6])$
```

Por conveniencia, para más adelante hacer un gráfico, convertimos la matrix datos en una lista. Esto es sencillo si utilizamos el comando "args":

```
→ datosL: args(datos);
(datosL) [[1,2],[3,2],[4,5],[6,6]]
```

Supondremos entonces que los puntos se ajustan a un polinomio lineal del tipo: $y = ax$. El parámetro a se calcula con la función "lsquares estimates". Es importante prestar bastante atención a la sintaxis del siguiente comando.

```
→ param: lsquares_estimates(datos,[x,y],y=a·x,[a]), numer;
(param) [a=1.032258064516129]
```


Este será entonces el valor del parámetro a de la ecuación de la recta $y = ax$ que pasa por el origen. Notemos también que le hemos asignado el valor del parámetro a a la variable param. Lo que haremos ahora es escribir la ecuación de dicha recta. Podemos hacer uso de la instrucción "ev" que nos permite evaluar una expresión.

```
→ y:ev(a·x,first(param));
(y) 1.032258064516129 x
```

Procederemos ahora a graficar los datos experimentales vs el ajuste por mínimos cuadrados en un mismo gráfico. Recordemos que el conjunto de puntos lo tenemos en la forma de una lista, que hemos denominada más arriba como datosL. Mientras que al ajuste que hemos calculado, es decir, la recta: $1,032258064516129x$ le hemos asignado la variable denominada y.

Es recomendable consultar el manual del programa, en la parte que tiene que ver con gráficos, plot2d, plot3d, para identificar la sintaxis que aparecerá en la siguiente instrucción.

→ `wxplot2d([[discrete,datosL], y], [x,0,10],[style, [points,5,2], [lines,2,1]], [point_type,`



Nota: Se deja como ejercicio repetir éste mismo cálculo pero usando un ajuste para los datos de la forma: $y = ax + b$.

→ `kill(all)$`

2. Consideraremos el conjunto de datos:

$$|x1\rangle = (1, 2, 1, 1), |x2\rangle = (2, 1, 1, -1), |y\rangle = (15, 12, 10, 0).$$

Vamos a suponer que ajustan de la manera siguiente: $|y\rangle = a|x1\rangle + b|x2\rangle$.

→ `datos2: matrix([1,2,15],[2,1,12],[1,1,10],[1,-1,0])$`

Cambiemos ligeramente la notación por: $z = ax + by$ y calculemos los parámetros a y b.

```
→ param: lsquares_estimates(datos2,[x,y,z], z=a·x+b·y,[a,b]), numer;
(param) [[a=4.090909090909091, b=5.090909090909091]]
```

3. Para el tercer ejemplo se consideraron los siguientes datos:

$$\{(0,1),(1,3),(2,7),(3,15)\} \Leftrightarrow y=ax^2+bx+c.$$

Haremos con Maxima el cálculo directo usando un ajuste cuadrático para los datos suministrados.

```
→ datos3: matrix([0,1],[1,3],[2,7],[3,15])$  

→ datosL3: args(datos3)$  


→ param: lsquares_estimates(datos3,[x,y], y=a·x^2+b·x+c,[a,b,c]), numer;  

(param) [[a=1.5, b=0.1, c=1.1]]  

→ y2:ev(a·x^2+b·x+c,first(param))$
```

Como hicimos anteriormente, graficamos los datos y el ajuste cuadrático en una misma figura.

```
→ wxplot2d([[discrete,datosL3], y2], [x,0,4],[style, [points,5,2], [lines,2,1]], [point_type,
```



```
→ kill(all)$
```

3.- Polinomios ortogonales

Maxima contiene la librería "orthopoly" que nos permite acceder a la evaluación simbólica y numérica de los diferentes tipos de polinomios ortogonales: Chebyshev, Laguerre, Hermite, Jacobi, Legendre...

→ `load (orthopoly)$`

Por ejemplo, para obtener los primeros 6 polinomios de Legendre escribimos los siguientes comandos:

→ `[legendre_p(0,x),legendre_p(1,x),legendre_p(2,x),legendre_p(3,x),legendre_p(4,x),`


Simplificamos con "ratsimp":

→ `ratsimp (%);`

$$(\%o3) \quad [1, x, \frac{3x^2 - 1}{2}, \frac{5x^3 - 3x}{2}, \frac{35x^4 - 30x^2 + 3}{8}, \\ \frac{63x^5 - 70x^3 + 15x}{8}]$$

Los diferentes polinomios de Legendre se pueden visualizar de la manera siguiente:

→ `wxplot2d(%,[x,-1,1],[legend, "P0", "P1","P2", "P3","P4", "P5"])$`



Ahora bien, con los datos de la figura 2.3 se planteó un sistema de ecuaciones lineales:

→ $\text{ecu1:C0}-\text{C1}+\text{C2}-\text{C3}+\text{C4}-\text{C5}=8\$$

→ $\text{ecu2:C0}-3/5\cdot\text{C1}+1/25\cdot\text{C2}+9/25\cdot\text{C3}-51/125\cdot\text{C4}+477/3125\cdot\text{C5}=10\$$

→ $\text{ecu3:C0}-1/5\cdot\text{C1}-11/25\cdot\text{C2}+7/25\cdot\text{C3}+29/125\cdot\text{C4}-961/3125\cdot\text{C5}=11\$$

→ $\text{ecu4:C0}+1/5\cdot\text{C1}-11/25\cdot\text{C2}-7/25\cdot\text{C3}+29/125\cdot\text{C4}+961/3125\cdot\text{C5}=18\$$

→ $\text{ecu5:C0}+3/5\cdot\text{C1}+1/25\cdot\text{C2}-9/25\cdot\text{C3}-51/125\cdot\text{C4}-477/3125\cdot\text{C5}=20\$$

→ $\text{ecu6:C0}+\text{C1}+\text{C2}+\text{C3}+\text{C4}+\text{C5}=34\$$

Resolvemos el sistema anterior:

→ $\text{linsolve}([\text{ecu1}, \text{ecu2}, \text{ecu3}, \text{ecu4}, \text{ecu5}, \text{ecu6}], [\text{C0}, \text{C1}, \text{C2}, \text{C3}, \text{C4}, \text{C5}]);$

(%o11) $[\text{C0} = \frac{2249}{144}, \text{C1} = \frac{3043}{336}, \text{C2} = \frac{1775}{504}, \text{C3} = -\frac{175}{216}, \text{C4} = \frac{625}{336}, \text{C5}$
 $= \frac{14375}{3024}]$

Para asignar cada solución a la variable correspondiente podemos hacer lo siguiente:

→ $[\text{C0}, \text{C1}, \text{C2}, \text{C3}, \text{C4}, \text{C5}]:[\text{rhs}(\%), \text{rhs}(\%), \text{rhs}(\%), \text{rhs}(\%), \text{rhs}(\%), \text{rhs}(\%)];$

(%o12) $[\frac{2249}{144}, \frac{3043}{336}, \frac{1775}{504}, -\frac{175}{216}, \frac{625}{336}, \frac{14375}{3024}]$

Por lo tanto, la función aproximada será:

→ $f:\text{C0}+\text{C1}\cdot\text{legendre_p}(1, x)+\text{C2}\cdot\text{legendre_p}(2, x)+\text{C3}\cdot\text{legendre_p}(3, x)+\text{C4}\cdot\text{legendre_p}(4, x)$

→ $f:\text{expand}(%);$

(f) $\frac{14375 x^5}{384} + \frac{3125 x^4}{384} - \frac{8375 x^3}{192} - \frac{325 x^2}{192} + \frac{7367 x}{384} + \frac{1863}{128}$

Procedemos a introducir los datos:

→ $\text{datos}:[[-1, 8], [-3/5, 10], [-1/5, 11], [1/5, 18], [3/5, 20], [1, 34]];$


(datos) $[[-1, 8], [-\frac{3}{5}, 10], [-\frac{1}{5}, 11], [\frac{1}{5}, 18], [\frac{3}{5}, 20], [1, 34]]$

Para finalizar, haremos la gráfica con los datos y con la interpolación:

→

```
wxplot2d([[discrete,datos],f], [x,-1,1],[style, [points,5,2], [lines,2,1]],[point_type, p
```

(%t16)



→

```
kill(all)$
```

FIN